

```
-- file BootControl.Mesa
-- last edited by Sandman, August 31, 1978 5:23 PM

DIRECTORY
  BootCacheDefs: FROM "bootcachedefs",
  BootmesaDefs: FROM "bootmesadefs",
  CommanderDefs: FROM "commanderdefs",
  ImageDefs: FROM "imagedefs",
  MiscDefs: FROM "miscdefs",
  StringDefs: FROM "stringdefs",
  TimeDefs: FROM "timedefs";

BootControl: PROGRAM
  IMPORTS BootCacheDefs, BootmesaDefs, CommanderDefs, ImageDefs, StringDefs,
    TimeDefs, MiscDefs =

BEGIN

Data: BootmesaDefs.BootData;
imagenam: STRING ← [40];

WriteImage: PROCEDURE =
  BEGIN OPEN TimeDefs;
  StringDefs.AppendString[herald,"Alto/Mesa Bootmesa "L];
  AppendDayTime[herald,UnpackDT[TimeDefs.CurrentDayTime[]]];
  herald.length ← herald.length-3; -- remove seconds
  ImageDefs.MakeImage["Bootmesa.Image."L];
  RETURN
  END;

RunInitCode: PROCEDURE =
  BEGIN
  MiscDefs.SetBlock[@Data, NIL, SIZE[BootmesaDefs.BootData]];
  Data.imageFileRoot ← imagenam;
  START CommanderDefs.Commander[herald];
  START BootmesaDefs.Bootmesa[@Data];
  START BootmesaDefs.BootImage[@Data];
  START BootmesaDefs.BootScript[@Data];
  START BootCacheDefs.BootCache;
  START BootmesaDefs.BuildNucleus;
  END;

herald: STRING ← [40];

-- Main Body

RunInitCode[];
WriteImage[];
RESTART CommanderDefs.Commander;
END...
```